#######################################################################
# SConscript for xlib winsys

Import('*')

env = env.Clone()

env.Prepend(LIBS = [
    ws_xlib,
    mesautil,
    gallium,
])

env.Append(LIBS = env['X11_LIBS'])
env.Append(LIBPATH = env['X11_LIBPATH'])

env.Append(CPPPATH = [
    '#src/gallium/drivers',
    '#src/gallium/include/state_tracker',
    '#src/gallium/winsys',
])

sources = [
    'graw_xlib.c',
    graw_util
]

if True:
    env.Append(CPPDEFINES = ['GALLIUM_SOFTPIPE'])
    env.Prepend(LIBS = [softpipe])

if env['llvm']:
    env.Append(CPPDEFINES = 'GALLIUM_LLVMPIPE')
    env.Prepend(LIBS = [llvmpipe])

graw = env.SharedLibrary(
    target ='graw',
    source = sources,
)

graw = env.InstallSharedLibrary(graw, version=(1, 0))

env.Alias('graw-xlib', graw)
